
pkgname=libgcrypt
pkgver=1.8.3
pkgrel=1
pkgdesc="GNU's basic cryptographic library"
arch=('x86_64')
url="https://www.gnupg.org"
license=('LGPL')
depends=('libgpg-error')
options=('!libtool' 'emptydirs')
source=("https://www.gnupg.org/ftp/gcrypt/libgcrypt/${pkgname}-${pkgver}.tar.bz2")
sha1sums=('13bd2ce69e59ab538e959911dfae80ea309636e3')

build() {
  cd ${pkgname}-${pkgver}

  ./configure --prefix=/usr \
    --disable-padlock-support \
    --disable-static
  make
}

check() {
  cd ${pkgname}-${pkgver}
  
  make check
}

package() {
  cd ${pkgname}-${pkgver}

  make DESTDIR=${pkgdir} install
}
